

FreeFem++ の紹介

桂田 祐史

2007年9月26日～, 2016年2月14日, 2017年7月16日

この文書の最新版は

<http://nalab.mind.meiji.ac.jp/~mk/labo/text/welcome-to-freefem/>

または

<http://nalab.mind.meiji.ac.jp/~mk/labo/text/welcome-to-freefem.pdf>

で閲覧可能です。

FreeFem++ の文法については、(たくさんの例がついている) マニュアルを読むのが本筋でしょうけれど、「FreeFem++ノート」¹ というメモも用意してあります。

1 手短な紹介

偏微分方程式の数値解法には、古くから差分法 (finite difference method, FDM) と有限要素法 (finite element method, FEM) という二大手法があります (最近では有限体積法 (finite volume method) も良く使われるようになっていて、「二大」は止めた方が良いでしょう)。

有限要素法は、解析学の世界でスタンダードとなっている弱解の方法を数値計算に応用したもので、微分方程式の弱定式化の議論とほぼ並行した形で近似解の議論がなされ、数学者になじみやすいものです。また有限要素法は、プログラムの自動生成がしやすいという特徴を持っています²。このため、従来から、プログラムを自動生成するシステムを開発する試みが色々なされてきました。

FreeFem++³ は、パリ第6大学 J. L. Lions 研究所の Frédéric Hecht, Oliver Pironneau, A. Le Hyaric, 広島国際学院大学の 大塚厚二氏らが開発した、2次元, 3次元問題を有限要素法で解くための、一種の PSE (problem solving environment) で、<http://www.freefem.org/ff++/> からソースコード、マニュアル (400 ページ超, 幸い英文)、主なプラットフォーム (Windows, Mac, Linux) 向けの実行形式パッケージが公開されています。

FreeFem++ では、ユーザーが境界曲線を指定することで定義した2次元領域に対して、自動的に三角形要素分割を生成し、弱形式により記述された問題に対して、種々の有限要素 (関数) 空間を用いて弱解を求め、可視化することが可能になっています (最近では3次元領域の問題も解けるようになってきているようですが、筆者はそれについてほとんど知らないので、言及するのを止めておきます。)

問題の領域や弱形式などは専用の言語で記述した短いプログラム (ものによっては 20~30 行程度) として与え、それを実行することでシミュレーションを行います。

¹<http://nalab.mind.meiji.ac.jp/~mk/labo/text/freefem-note.pdf>

²こう書くと差分法のシンパに反発されそうな気もしますが、野次馬の観察 (岡目八目) によると、有限要素法に一日の長があるのは否定できないと思います。

³<http://www.freefem.org/ff++/>

すべての問題がこのソフトで扱えるわけではありませんが、扱える問題は結構幅広く、その場合は、C言語のようなプログラミング言語で一からプログラムを書くのに比べて、桁違いに短い時間でシミュレーション実行までたどり着けます。(プログラムの行数自体が2桁とは行きませんが、 $1.x$ 桁小さいように思います。)

大塚・高石「有限要素法で学ぶ現象と数理—FreeFem++」共立出版(2014)という解説書が出版され、そのサポートサイト「有限要素法で学ぶ現象と数理」⁴が出来ました。有益な情報が日本語で得られるようになりました。

応用数理学会でFreeFem++のセミナーが何回か開かれています。私が参加できた直近2回の情報は

- ソフトウェアセミナー:FreeFem++による有限要素プログラミング—上級編—⁵(2016/6/4,5)
資料<https://www.ljll.math.upmc.fr/~suzukia/FreeFempp-tutorial-JSIAM2016b/index.html>
- ソフトウェアセミナー:FreeFem++で学ぶ現象と数理⁶(2016/2/11, 12)
資料<http://www.ljll.math.upmc.fr/~suzukia/FreeFempp-tutorial-JSIAM2016>

幸い、2回とも都合がついて参加でき、有益な情報が得られました。関係者のご尽力に感謝いたします。

講師の鈴木厚先生には、2007年の九州大学で行われたチュートリアルでもお世話になりました。

2 百聞は一見に如かず

2.1 Mac に最新版をインストールする

FreeFem++ 本体をインストールするには、WWW サイト Freefem++ Home Page⁷ から、FreeFem++-3.50-MacOS_10.11.pkg (3.50 はバージョン番号でこれは変化する) のようなインストーラーを入手して実行し (Ctrl+クリックして「開く」)、後は適当に PATH の設定をするだけです。

PATH は以前は /usr/local/bin で良かったですが⁸、最近は /usr/local/ff++/ の下にファイルを納めるようになったので、/usr/local/ff++/openmpi/3.50/bin (3.50 の場合) と /usr/local/ff++/openmpi/bin です。

bash を使っている場合 .profile などにこう書く

```
export PATH=$PATH:/usr/local/ff++/openmpi/3.50/bin:/usr/local/ff++/openmpi/bin
```

400 ページ超のマニュアルの PDF ファイル⁹、その中で紹介されているサンプル・プログラム¹⁰も得られます。

⁴<http://comfos.org/jp/ffempp/book/>

⁵<http://na.cs.tsukuba.ac.jp/acmi/?p=403>

⁶<http://na.cs.tsukuba.ac.jp/acmi/?p=383>

⁷<http://www.freefem.org/ff++/>

⁸ものは考えようで、以前は /usr/local/bin の下に色々インストールして、バージョンアップの際の整理がやりにくかった、とも言えますね。

⁹ /usr/local/ff++/openmpi/3.50/share/freefem++/freefem++doc.pdf

¹⁰ /usr/local/ff++/openmpi/3.50/share/freefem++/3.50/

GUI でプログラムの開発&実行の出来る **FreeFem++-cs** という IDE (統合開発環境) があって、以前は私も身の回りの学生に推奨していましたが、最近ではオリジナルの FreeFem++ だけで十分と考えるようになりました¹¹。

2.2 簡単な問題で数学超特急

Poisson 方程式の境界値問題で説明します (有限要素法を自習したい場合には、菊地文雄, 『有限要素法概説』, サイエンス社, という参考書がイチオシです, 桂田研の学生で希望する人は申し出て下さい)。

\mathbb{R}^2 内の領域 (連結開集合) Ω における Poisson 方程式の Dirichlet 問題

$$\begin{aligned} \text{(PE)} \quad & -\Delta u = f \quad (\text{in } \Omega), \\ \text{(DBC)} \quad & u = 0 \quad (\text{on } \Gamma := \partial\Omega) \end{aligned}$$

を考える (弱解の方法の例題として、もっとも簡単な、定番の問題です)。

Poisson 方程式に $v \in C_0^\infty(\Omega)$ ($C_0^\infty(\Omega)$ は、 Ω 内に compact な台を持つ C^∞ 級の関数全体) をかけて、Green の積分公式 (多次元版部分積分) を用いると、次式が得られます。

$$(1) \quad \iint_{\Omega} \left(\frac{\partial u}{\partial x} \frac{\partial v}{\partial x} + \frac{\partial u}{\partial y} \frac{\partial v}{\partial y} \right) dx dy = \iint_{\Omega} f(x, y)v(x, y) dx dy \quad (v \in C_0^\infty(\Omega)).$$

逆に $u = 0$ (on Γ) を満たす滑らかな u が、この条件を満たせば、 u は Poisson 方程式を満たすことも容易に分かります。

微分方程式を満たす u を求める代りに、(1) を満たす u を見出すことを考えましょう。解の存在を容易に保証できるようにするため、関数空間を完備化するのが便利です (解を極限の論法で得よう、という解析学の常套手段)。具体的には、以下に定義する $H_0^1(\Omega)$ というソボレフ空間 (一般化された導関数を持つ関数からなる関数空間の一種) で u を探すことにします。

(PE),(DBC) の弱解の定義

$u \in H_0^1(\Omega)$ かつ

$$\iint_{\Omega} \left(\frac{\partial u}{\partial x} \frac{\partial v}{\partial x} + \frac{\partial u}{\partial y} \frac{\partial v}{\partial y} \right) dx dy = \iint_{\Omega} f(x, y)v(x, y) dx dy \quad (v \in H_0^1(\Omega)).$$

$H_0^1(\Omega)$ とは

$$H^1(\Omega) = \left\{ u \in L^2(\Omega); \frac{\partial u}{\partial x_j} \in L^2(\Omega) \ (j = 1, 2) \right\}$$

という関数の集合に

$$(u, v)_{H^1} := \iint_{\Omega} u(x, y)\overline{v(x, y)} dx dy + \iint_{\Omega} \nabla u(x, y) \cdot \overline{\nabla v(x, y)} dx dy$$

という内積を導入すると、 $H^1(\Omega)$ は Hilbert 空間 (完備な内積空間) になる。

$$H_0^1(\Omega) := C_0^\infty(\Omega) \text{ の } H^1(\Omega) \text{ での閉包}$$

とおく。 $H_0^1(\Omega)$ の要素は、 $H^1(\Omega)$ の要素のうちで、ある意味で Γ 上 0 となるものとみなすことができる。

¹¹ というか、最近 Mac 用の FreeFem++-cs ってちゃんと動くんでしょうか？

Ω を三角形の「整った」合併 T_h で近似し、 T_h で連続で、各三角形上で1次関数であるような関数全体を V_h とします。 V_h は $H^1(\Omega)$ の有限次元部分空間とみなせ、グラフが三角錐であるような関数 $\varphi_1, \dots, \varphi_m$ を基底に取って、各元をその基底の線型結合として表現することができます:

$$V_h = \left\{ \sum_{j=1}^m c_j \varphi_j; (c_j) \in \mathbb{R}^m \right\}.$$

(PE),(DBC) の弱解の定義 (書き換え)

$u \in H^1(\Omega)$ かつ

$$\iint_{\Omega} \left(\frac{\partial u}{\partial x} \frac{\partial v}{\partial x} + \frac{\partial u}{\partial y} \frac{\partial v}{\partial y} \right) dx dy = \iint_{\Omega} f(x, y) v(x, y) dx dy \quad (v \in H^1(\Omega), v = 0 \text{ on } \Gamma),$$

$$u = 0 \quad (\text{on } \Gamma).$$

有限要素解の定義

$u_h \in V_h$ かつ

$$\iint_{T_h} \left(\frac{\partial u_h}{\partial x} \frac{\partial v_h}{\partial x} + \frac{\partial u_h}{\partial y} \frac{\partial v_h}{\partial y} \right) dx dy = \iint_{T_h} f(x, y) v_h(x, y) dx dy \quad (v_h \in V_h, v_h = 0 \text{ on } \Gamma),$$

$$u_h = 0 \quad (\text{on } \Gamma).$$

次のようなプログラム `poisson.edp` を用意します (テキスト・エディターを使って作成します。Mac だったら、OS 付属のテキストエディット¹² が使えます。現象数理学科 Mac だったら `mi` がお勧めかも。)

`poisson.edp`¹³

```
// 境界の定義 (単位円), いわゆる正の向き
border Gamma(t=0,2*pi) { x=cos(t); y=sin(t); }
// 三角形要素分割を生成 (境界を 50 に分割)
mesh Th = buildmesh(Gamma(50));
plot(Th,wait=true);
// 有限要素空間は P1 (区分的1次多項式) 要素
real [int] levels =-0.012:0.001:0.012;
fespace Vh(Th,P1);
Vh u,v;
// Poisson 方程式  $-\Delta u=f$  の右辺
func f = x*y;
// 現在時刻をメモ
real start = clock();
// 問題を解く
solve Poisson(u,v)
= int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th)(f*v)
+on(Gamma,u=0);
// 可視化 (等高線)
plot(u,wait=true);
//plot(u,viso=levels,fill=true,wait=true);
// 可視化 (3次元) --- マウスで使って動かせる
//plot(u,dim=3,wait=true);
plot(u,dim=3,viso=levels,fill=true,wait=true);
// 計算時間を表示
cout << " CPU time= " << clock() - start << endl;
```

¹²<http://nalab.mind.meiji.ac.jp/~mk/knowhow-2015/node1.html>

FreeFem++ で poisson.edp を実行

```
mymac% FreeFem++ poisson.edp
```

キーボードからのコマンド

| | |
|---------|------------------|
| [Enter] | 次のプロットへ |
| [ESC] | グラフィックスを閉じる |
| + | ズーム (イン) する |
| - | ズームアウトする |
| = | ズームの状態を元に戻す |
| c | ベクトルの矢印の大きさを短くする |
| C | ベクトルの矢印の大きさを長くする |
| r | リフレッシュする |
| f | カラーを塗りつぶすかどうか |
| v | |
| ? | help |

`plot(u);` のところを `plot(u,ps="graph.eps");` のようにすると、画面に表示するだけでなく、PostScript 形式でファイル出力できます (TEX 文書への取り込みが簡単です)。




図 1: 円盤領域で Poisson 方程式 $-\Delta u(x, y) = xy$ の同次 Dirichlet 問題を解く (要素分割と解の等高線)




図 2: dim=3 でグラフの鳥瞰図を描く

3 おまけ: gnuplot で可視化

従来、FreeFem++ では、等高線描画やベクトル場の表示などは出来るが、グラフの鳥瞰図描画などはサポートしていなかった(今は `plot(...,dim=3,...)`; とするだけで出来る)。そこで gnuplot を使ってグラフを描く方法を紹介する。

FreeFem++ プログラム `poisson-disk.edp`

```
// 境界の定義 (単位円), いわゆる正の向き
border Gamma(t=0,2*pi) { x=cos(t); y=sin(t); }
// 三角形要素分割を生成 (境界を 50 に分割)
mesh Th = buildmesh(Gamma(50));
plot(Th,ps="mesh.eps");
plot(Th,wait=1);
// 有限要素空間は P1 (区分的1次多項式) 要素
fespace Vh(Th,P1);
Vh u,v;
// Poisson 方程式  $-\Delta u=f$  の右辺
func f = x*y;
// 現在時刻をメモ
real start = clock();
// 問題を解く
solve Poisson(u,v)
  = int2d(Th) (dx(u)*dx(v)+dy(u)*dy(v))-int2d(Th) (f*v)
  +on(Gamma,u=0);
// 可視化
plot(u,ps="poisson-disk.eps");
plot(u);
//
{
  ofstream ug("u-g.txt");
  for (int i=0; i<Th.nt; i++) {
    for (int j=0; j<3; j++) {
      ug << Th[i][j].x << " " << Th[i][j].y << " " << u[][Vh(i,j)]<<endl;
    }
    ug << Th[i][0].x << " " << Th[i][0].y << " " << u[][Vh(i,0)]<<"\n\n\n";
  }
}
// 計算時間を表示
cout << " CPU time= " << clock() - start << endl;
```

“u-g.txt” は次のようなデータである (3次元空間における三角形が並んでいる)。

```
0.735953 0.529892 0.00584987
0.876307 0.481754 6.33978e-033
0.809017 0.587785 7.53444e-033
0.735953 0.529892 0.00584987
```

```
0.512882 0.46938 0.0104192
0.577982 0.366799 0.00923669
0.656968 0.448345 0.00912418
0.512882 0.46938 0.0104192
```

(中略)

.....

```
0.656968 0.448345 0.00912418
0.447782 0.737755 0.00678195
0.425779 0.904827 4.83201e-033
0.317582 0.777124 0.00599877
0.447782 0.737755 0.00678195
```

```
0.187381 0.982287 3.00659e-033
0.0627905 0.998027 8.0753e-034
0.149059 0.864343 0.00242012
0.187381 0.982287 3.00659e-033
```

このデータを gnuplot で可視化するには、例えば次のようにする (図 3)。

```
poisson-disk.g
set hidden3d
set palette rgbformulae 33,13,10
splot "u-g.txt" with lines palette
```




図 3: Poisson 方程式の解のグラフ表示 (gnuplot による)

参考文献

- [1] 菊地文雄：有限要素法概説，サイエンス社 (1980)，新訂版 1999。

A 参考: FreeFem++ で菊地 [1] の Poisson 方程式の例題を解く

菊地 [1] に載っている Poisson 方程式の例題

$$\begin{aligned} (2) \quad & -\Delta u = f \quad \text{in } \Omega, \\ (3) \quad & u = g_1 \quad \text{in } \Gamma_1, \\ (4) \quad & \frac{\partial u}{\partial n} = g_2 \quad \text{in } \Gamma_2 \end{aligned}$$

(ただし、 $\Omega = (0, 1) \times (0, 1)$, $\Gamma_1 = \{0\} \times [0, 1] \cup [0, 1] \times \{0\}$, $\Gamma_2 = \{1\} \times (0, 1] \cup (0, 1] \times \{1\}$, $f = 1$, $g_1 = 0$, $g_2 = 0$) を FreeFem++ を用いて解くとどうなるか。

次のようなプログラムで解ける。

```
Poisson2.edp
// Poisson2.edp
int Gamma1=1, Gamma2=2;
border Gamma10(t=0,1) { x=0; y=1-t; label=Gamma1; }
border Gamma11(t=0,1) { x=t; y=0; label=Gamma1; }
border Gamma20(t=0,1) { x=1; y=t; label=Gamma2; }
border Gamma21(t=0,1) { x=1-t; y=1; label=Gamma2; }
int m=10;
mesh Th = buildmesh(Gamma10(m)+Gamma11(m)+Gamma20(m)+Gamma21(m));
plot(Th, wait=1,ps="Th.eps");
savemesh(Th,"Th.msh"); // optional
fespace Vh(Th,P1);
Vh u,v;
func f=1;
func g1=0;
func g2=0;
solve Poisson(u,v) =
  int2d(Th)(dx(u)*dx(v)+dy(u)*dy(v))
  -int2d(Th)(f*v)
  -int1d(Th,Gamma2)(g2*v)
  +on(Gamma1,u=g1); // on(Gamma10,Gamma11,u=g1)
plot(u,ps="contour.eps");
```

§2.2 のサンプル・プログラムを叩き台にする。

プログラムで生成される、三角形要素分割のデータを出力できるが、それは次のようなものである (しばしば、数値実験の三角形要素分割を、FreeFem++ の出力を拝借している研究者がいる)。




図 4: Poisson2.edp — 要素分割と解の等高線

$m = 2$ のときの “Th.msh”

```

9 8 8
0 0 1
0 1 2
0 0.5 1
0.5 0 1
1 0 2
0.4999999999488 0.4999999999488 0
0.5 1 2
1 0.5 2
1 1 2
7 8 9 0
1 4 3 0
4 5 8 0
6 8 7 0
4 6 3 0
4 8 6 0
2 3 7 0
7 3 6 0
9 7 2
7 2 2
5 8 2
8 9 2
1 4 1
4 5 1
2 3 1
3 1 1

```

マニュアルで確認していないが、多分次のようなフォーマットになっているのであろう。




図 5: $m = 2$ の時の三角形分割の様子 (Th.msh に対応)

総節点数 n 総要素数 m 境界に属する辺の数 N

節点 P_1 の x, y 座標とラベル (0 は内点)

⋮

節点 P_n の x, y 座標とラベル (0 は内点)

要素 e_1 の 3 節点の節点番号と 0

要素 e_2 の 3 節点の節点番号と 0

⋮

要素 e_m の 3 節点の節点番号と 0

境界に属する辺 Q_1 とラベル

⋮

境界に属する辺 Q_N とラベル

B 2次元 Laplacian の固有値問題を FreeFem++ で解く

(ここは書きなおす必要があることが判明したけれど、しばらくは実行する時間が取れない…書いてあることを信用しないように。)

F. Hecht, FreeFem++ a software to solve PDE¹⁴ に分かりやすい解説がある。次に掲げるプログラムはそこに載っているものである。

¹⁴<http://www.freefem.org/ff++/ftp/CIMPA/CIMPA-Guadeloupe-FF.pdf>

```

verbosity=10 ;
mesh Th=square(20,20,[pi*x,pi*y]);
fespace Vh(Th,P2);
Vh u1,u2;
real sigma = 20; // value of the shift
// OP = A - sigma B; // the shifted matrix
varf op(u1,u2)= int2d(Th)( dx(u1)*dx(u2) + dy(u1)*dy(u2) - sigma* u1*u2 )
+ on(1,2,3,4,u1=0) ; // Boundary condition
varf b([u1],[u2]) = int2d(Th)( u1*u2 ); // no Boundary condition
matrix OP= op(Vh,Vh,solver=CROUT,factorize=1);
// crout solver because the matrix is not positive
matrix B= b(Vh,Vh,solver=CG,eps=1e-20);
// important remark:
// the boundary condition is made with exact penalisation:
// we put 1e30=10^30 on the diagonal term to lock the degree of freedom.
// So take Dirichlet boundary condition just on a variational form
// and not on the b variational form.
// because we solve w=OP^-1*B*v


int nev=20; // number of computed eigenvalues close to sigma
real[int] ev(nev); // to store the nev eigenvalues
Vh[int] eV(nev); // to store the nev eigenvectors

int k=EigenValue(OP,B,sym=true,sigma=sigma,value=ev,vector=eV,
tol=1e-10,maxit=0,ncv=0) ;
// return the number of computed eigenvalues

for (int i=0;i<k;i++) {
u1=eV[i];
real gg = int2d(Th)(dx(u1)*dx(u1) + dy(u1)*dy(u1));
real mm= int2d(Th)(u1*u1);
cout<<"----"<< i<<" "<<ev[i]<<"err="
<<int2d(Th)(dx(u1)*dx(u1) + dy(u1)*dy(u1) - (ev[i])*u1*u1)
<< " --- " <<endl;
plot(eV[i],cmm="Eigen Vector "+i+" valeur =" + ev[i] ,wait=1,value=1);
}

```

リンク切れ…と思ったら、マニュアルに入った (2015/12/22 の時点で §9.4)。



なお、8行目の `+on(1,2,3,4,u=0)` を削除すると、Dirichlet 境界条件の代わりに、自然境界条件である Neumann 境界条件となる。

$$-\Delta u = \lambda u \quad \text{in } \Omega, \quad u = 0 \quad (\text{on } \partial\Omega)$$

の両辺に試験関数 v をかけて積分し、部分積分すると、

$$\iint_{\Omega} \nabla u \cdot \nabla v \, dx \, dy = \lambda \iint_{\Omega} uv \, dx \, dy.$$

すなわち

$$\iint_{\Omega} (u_x v_x + u_y v_y) \, dx \, dy = \lambda \iint_{\Omega} uv \, dx \, dy.$$

これを有限要素法で離散化すると、一般化固有値問題と呼ばれる

$$A\mathbf{u} = \lambda B\mathbf{u}$$

という形をした方程式が得られる。ここで B は正値対称、 A は実対称である。

「シフト法で解く」と簡単に書いてあるだけで、詳しいことは書いてない。次のようなことかと推測する。一般化固有値に近いと考えられる実数 σ を与えたとき、 $A\mathbf{u} = \lambda B\mathbf{u}$ の両辺から $\sigma B\mathbf{u}$ を引くと、

$$(A - \sigma B)\mathbf{u} = (\lambda - \sigma)B\mathbf{u}.$$

ゆえに

$$(A - \sigma B)^{-1} B\mathbf{u} = \frac{1}{\lambda - \sigma} \mathbf{u}.$$

これは \mathbf{u} が、行列 $(A - \sigma B)^{-1} B$ の、固有値 $\frac{1}{\lambda - \sigma}$ に属する固有ベクトルであることを示す。ゆえに行列 $(A - \sigma B)^{-1} B$ について冪乗法を実行すれば、絶対値が最大であるような固有値 $\mu = \frac{1}{\lambda - \sigma}$ が得られる。

細かいことだが、重要な注意がある。ここでは square() を用いてメッシュ分割をしているが、それを用いずに、自分で境界曲線を定義してメッシュ分割をすると、真の固有値が重複 (縮重) する場合であっても、数値計算で得られる近似固有値は重複 (縮重) しない。これは自動メッシュ分割で得られる三角形分割は、正方形の二面体群 (合同変換全体のなす群) D_4 の対称性を持たないため、近似固有値が重複しないためである。

C FreeFem++ のマニュアル §3.9 から: Stokes 方程式の cavity flow

```
// stokes.edp (in the manual, chapter 3)
int n=3;
mesh Th=square(10*n,10*n);
plot(Th, wait=1, ps="stokes-p1b-mesh.eps");
fespace Uh(Th,P1b); Uh u,v,uu,vv;
fespace Ph(Th,P1); Ph p,pp;
solve stokes([u,v,p],[uu,vv,pp]) =
int2d(Th)(dx(u)*dx(uu)+dy(u)*dy(uu) + dx(v)*dx(vv)+ dy(v)*dy(vv)
+ dx(p)*uu + dy(p)*vv + pp*(dx(u)+dy(v)))
+ on(1,2,4,u=0,v=0) + on(3,u=1,v=0);
plot([u,v],p,wait=1,ps="stokes-p1b-solution.eps");
```

$$\mathbf{u}_h = \begin{pmatrix} u_1 \\ u_2 \end{pmatrix} \leftarrow \begin{pmatrix} \mathbf{u} \\ \mathbf{v} \end{pmatrix}, \quad \mathbf{v}_h = \begin{pmatrix} v_1 \\ v_2 \end{pmatrix} \leftarrow \begin{pmatrix} \mathbf{uu} \\ \mathbf{vv} \end{pmatrix}, \quad p_h \leftarrow p, \quad q_h \leftarrow pp, \quad x_1 \leftarrow x, \quad x_2 \leftarrow y$$

とすると、

$$\iint_{\Omega_h} \left(\frac{\partial u_1}{\partial x_1} \frac{\partial v_1}{\partial x_1} + \frac{\partial u_1}{\partial x_2} \frac{\partial v_1}{\partial x_2} + \frac{\partial u_2}{\partial x_1} \frac{\partial v_2}{\partial x_1} + \frac{\partial u_2}{\partial x_2} \frac{\partial v_2}{\partial x_2} \right) dx dy$$

$$+ \iint_{\Omega_h} \left(\frac{\partial p_h}{\partial x_1} v_1 + \frac{\partial p_h}{\partial x_2} v_2 \right) dx dy + \iint_{\Omega_h} q_h \left(\frac{\partial u_1}{\partial x_1} + \frac{\partial u_2}{\partial x_2} \right) dx dy = 0,$$

すなわち

$$a(\mathbf{u}_h, \mathbf{v}_h) + (\nabla p_h, \mathbf{v}_h) + (q_h, \nabla \cdot \mathbf{u}_h) = 0 \quad (\forall (\mathbf{v}_h, q_h)).$$

これは

$$\begin{cases} a(\mathbf{u}_h, \mathbf{v}_h) + (\nabla p_h, \mathbf{v}_h) = 0 & (\forall \mathbf{v}_h), \\ (q_h, \nabla \cdot \mathbf{u}_h) = 0 & (\forall q_h) \end{cases}$$

と同値である。

第2項が $-(p_h, \nabla \cdot \mathbf{v}_h)$ でないところが目につく。そうするためには

```
int2d(Th) (dx(u)*dx(uu)+dy(u)*dy(uu) + dx(v)*dx(vv)+ dy(v)*dy(vv)
-p*(dx(uu)+dy(vv))+ dy(p)*vv + pp*(dx(u)+dy(v)))
+ on(1,2,4,u=0,v=0) + on(3,u=1,v=0);
```

とするわけだろう。

とにかく P1b/P1 を採用した結果は、図 6 となる。




図 6: stokes-p1b.edp の結果

P1/P1 にすると上手く計算できないことは常識とされているが、自分でプログラムを書く場合、なかなかそこまで試す気にはなれない。しかし FreeFem++ でそれをするのは簡単である (図 7)。

P2/P1 は、良く知られているようにきちんと解ける (図 8)。

他にも、悪い冗談のようだが P2/P2 などが気軽に試せる。

当たり前かもしれないが、弱形式を少しいじっただけで結果がコロコロ変わる。

D 2015年応用数学会チュートリアル

私事になりますが、初めて FreeFem++ を体験したのは、2007年春に九州大学で行われたチュートリアル (講師: 鈴木厚先生) がきっかけでした。




図 7: もしも P1/P1 とすると…なんだ、これは?!




図 8: P2/P1 (分割は粗くした)

そのとき頂いた資料をネタにして、自分の学生に FreeFem++ を教えて来ましたが、1 節に書いたように、2015, 2016 年に応用数学会の主催でチュートリアルが2回開かれ、講義資料が WWW で公開されています。今後はこの資料を使って学生指導をすることになると思います。